Math 1120 Maple Examples

```
> # This is a comment.
  # Let's clear memory and load the plots package so I can use the
  "display" command.
  restart;
  with(plots):
```

Example: Function vs. Expression

"->" means "maps to" so "x -> x^2;" means x maps to x^2. Then "f := x -> x^2;" means that I am assigning the mapping (that sends x to x^2) the name f. Finally, the commands I show below (1) define f (and suppress output) and (2) print out f(x) = x^2.

The first line is the one that we need. The second line is just to print out our function all nice and pretty.

```
> f := x -> x^2:
  'f(x)' = f(x);
```

$$f(x) = x^2 \tag{1}$$

Whereas, f is a function, g (defined below) is an expression.

```
> g := x^2;
```

$$g := x^2 \tag{2}$$

We can use function notation and plug stuff into functions. To plug stuff into expressions we must use the "subs" (short for substitute) command.

```
> f(2);
```

$$4 \tag{3}$$

```
> subs(x=2,g);
```

$$4 \tag{4}$$

```
> f(Frank);
```

$$Frank^2 \tag{5}$$

```
> subs(x=Frank,g);
```

$$Frank^2 \tag{6}$$

Sometimes functions are easier to use and sometimes expressions are a better choice. If you look through my Maple example pages, you'll see both!

Example: Arc Length, Plotting, Approximating, and Solving

Let's plot sine on the interval $I = [-\pi, 2\pi]$.

```
> plot(sin(x),x=-Pi..2*Pi);
```



Now let's compute the arc length of this part of the sine function (first exactly and then get a decimal approximation).

We note that "sqrt" is the square root function, "diff" differentiates, and "int" integrates.

I will use inert commands (capitalized versions) to display what I'm trying to calculate and set them equal to active commands that will actually get the job done.

```
> Int(sqrt(1+(Diff(sin(x),x))^2),x=-Pi..2*Pi) = int(sqrt(1+(diff(sin(x),x))
  ^2),x=-Pi..2*Pi);
```

$$\int_{-\pi}^{2\pi} \sqrt{1+\left(\frac{\mathrm{d}}{\mathrm{d}x}\sin(x)\right)^2}\,\mathrm{d}x = 6\sqrt{2}\,\mathrm{EllipticE}\left(\frac{\sqrt{2}}{2}\right) \tag{7}$$

The "evalf" (evaluate to floating point) command will give us a decimal approximation.

```
> evalf(int(sqrt(1+(diff(sin(x),x))^2),x=-Pi..2*Pi));
```

$$11.46059336 \tag{8}$$

Let's see how far along the curve we need to go until we've gone 5 units. In other words, we need to calculate the arc length from $x = -\pi$ up to some unknown $x = b$ and then solve for b. First, the exact solver "solve" chokes, so we try again and numerically solve using "fsolve".

```
> solve(int(sqrt(1+(diff(sin(x),x))^2),x=-Pi..b)=5);
Warning, unable to determine if Pi*_Z1 is between -Pi and b; try to use 
assumptions or use the AllSolutions option
Error, (in solve) cannot solve expressions with int((1+cos(x)^2)^(1/2),x = -
Pi .. b) for b
> fsolve(int(sqrt(1+(diff(sin(x),x))^2),x=-Pi..b)=5);
Warning, unable to determine if Pi*_Z5 is between -Pi and b; try to use 
assumptions or use the AllSolutions option
```

$$0.8864881477 \tag{9}$$

So if b is approximately 0.88649, then the arc length of sin(x) for $-\pi \le x \le 0.88649$ is about 5 units.

```
> Int(sqrt(1+Diff(sin(x),x)^2),x=-Pi..0.88649) = int(sqrt(1+diff(sin(x),x)
  ^2),x=-Pi..0.88649);
```

$$\int_{-\pi}^{0.88649} \sqrt{1+\left(\frac{d}{dx}\sin(x)\right)^2}\, dx = 5.000002191 \quad \textbf{(10)}$$

Let's plot sin(x) over our whole domain and then overlay (in thick blue) the part that accounts for 5 units of arc length.

Notice that I saved these plots under the names "plot1" and "plot2" while suppressing output (ending with colons) and then use the "display" command to show both plots together.

```
> plot1 := plot(sin(x),x=-Pi..2*Pi):
  plot2 := plot(sin(x),x=-Pi..0.88649,color=blue,thickness=5):
  display({plot1,plot2});
```



Example: Let's compute an improper integral.

I'll do the inert commands on the left hand side and active ones on the right so that we print out what we're computing set equal to the desired result.

```
> Int(1/(x^2+1),x=-infinity..infinity) = int(1/(x^2+1),x=-infinity..
  infinity);
```

$$\int_{-\infty}^{\infty} \frac{1}{x^2+1}\, dx = \pi \quad \textbf{(11)}$$

```
> plot(1/(x^2+1),x=-10..10);
```



Example: Suppose that we have a group of 500 women whose heights are distributed normally. Their mean height is 65 inches with a standard deviation of 4 inches.

Let's define the appropriate distribution function. One usually uses σ to denote the standard deviation and μ to denote the mean.

```
> n := x -> 1/(sigma*sqrt(2*Pi))*exp(-1/2*((x-mu)/sigma)^2):
  'n(x)' = n(x);

  mu := 65;
  sigma := 4;

  'n(x)' = simplify(n(x));
```

$$n(x) = \frac{\sqrt{2}\,\mathrm{e}^{-\frac{(x-\mu)^2}{2\sigma^2}}}{2\sigma\sqrt{\pi}}$$

$$\mu := 65$$

$$\sigma := 4$$

$$n(x) = \frac{\sqrt{2}\,\mathrm{e}^{-\frac{(x-65)^2}{32}}}{8\sqrt{\pi}} \qquad \textbf{(12)}$$

The integral of n(x) from x=a to b gives the probability that one of these women has a height between a and b inches. If we multiply this by 500, we should get (approximately) how many women from our group fall into the height range.

How many women are between 5'10" and 5'11" (that is 70 and 71 inches tall)?

```
> evalf(500*int(n(x),x=70..71));
```

$$19.4212863 \qquad \textbf{(13)}$$

We should expect that about 19 women are between 5'10" and 5'11".

How many women (from our group) are under 5' tall (under 60 inches tall)?

```
> evalf(500*int(n(x),x=-infinity..60));
```

$$52.8248869 \qquad \textbf{(14)}$$

About 53 women in our group are 5' tall or shorter.

How tall would one need to be to among the tallest 1% of women in our group?

Here we want to find a such that $\int_a^{\infty} n(x)\,\mathrm{d}x = 0.01$.

I will use "fsolve" to get a decimal approximation of my answer. Oddly, Maple has trouble solving this equation, so we have to help it. Notice the argument "a=65". This is a guess at what a might be. Of course, it's not a particularly good guess, but it is enough to get Maple to lock in on the correct number.

```
> fsolve(int(n(x),x=a..infinity)=0.01,a=65);
```

(15)

$$74.30539150 \tag{15}$$

This means that anyone in our group who is 74.3 inches (about 6'2") or taller is amongst the tallest 1% of our group.

Example: Let's find the hydrostatic force of water against the side wall of a cylindrical tank. Specifically, we want to find the hydrostatic force against a half-disk wall of radius 5 meters.

Let y be the distance from the bottom of the tank. Then since the edge of the tank is circular: $x^2 + (y-5)^2 = 5^2$, we have that $x = \pm\sqrt{25-(y-5)^2} = \pm\sqrt{10\,y-y^2}$ so the width of the tank at depth y is $\sqrt{10\,y-y^2} - \left(-\sqrt{10\,y-y^2}\right) = 2\sqrt{10\,y-y^2}$. Thus the area of a strip of area y meters from the bottom is $2\sqrt{10\,y-y^2}\;dy$ square meters. The weight of water is 9800 N per cubic meter, so the hydrostatic force on the strip of wall y meters from the bottom is $9800\cdot(5-y)\cdot 2\sqrt{10\,y-y^2}\;dy$ since this strip is $5-y$ meters from the top of the tank. If we "add" up the forces from bottom $(y=0)$ to top $(y=5)$ meters, we'll have the force on the entire tank wall.

$$\mathit{false}$$

$$2\sqrt{-y^2+10\,y} = 2\sqrt{-y^2+10\,y}$$

$$2\sqrt{-y^2+10\,y}\;dy \tag{16}$$

```
> Int(9800*(5-y)*2*sqrt(10*y-y^2),y=0..5) = evalf(int(9800*(5-y)*2*sqrt(10*
  y-y^2),y=0..5));
```

$$\int_0^5 19600\,(5-y)\sqrt{-y^2+10\,y}\;\mathrm{d}y = 816666.6667 \tag{17}$$

Thus there is about 816,667 N of hydrostatic force on this half-disk tank wall.

How full would the tank have to be so the force was just 100,000 N?

So now instead of filling to a depth of 5 meters, we fill to a depth of b meters and determine what that is so our integral equals 100,000. Notice that the depth of a strip is no longer $5-y$. It's now $b-y$.

```
> Int(9800*(b-y)*2*sqrt(10*y-y^2),y=0..b);
```

$$\int_0^b 19600\,(b-y)\sqrt{-y^2+10\,y}\;\mathrm{d}y \tag{18}$$

We want this to be 100,000 N.

```
> Int(9800*(b-y)*2*sqrt(10*y-y^2),y=0..b) = 100000;
```

$$\int_0^b 19600\,(b-y)\sqrt{-y^2+10\,y}\;\mathrm{d}y = 100000 \tag{19}$$

```
> fsolve(int(9800*(b-y)*2*sqrt(10*y-y^2),y=0..b) = 100000);
```

$$2.093821060 \tag{20}$$

So if we fill the tank up to a depth of about 2.09 meters, the hydrostatic force on the side of our tank will be 100, 000 N.

```
> with(plots):
  waterPlot := shadebetween(-5+2.09,-sqrt(25-x^2), x=-4.066..4.066, color=
  "Blue"):
  tankPlot := plot(-sqrt(25-x^2),x=-5..5,color=black,thickness=10):
  tankTop := plot(0,x=-5.125..5.125,color=black,thickness=10):
  display({waterPlot,tankPlot,tankTop},scaling=constrained,axes=none);
```



<u>Example:</u> Initial Condition? An indefinite integral gives an expression which accounts for all possible antiderivatives. In Maple, "int(f,x)" find a specific anti-derivative so "int(f,x) + C" (where C is an arbitrary constant) would be our indefinite integral.

What if we want to pick out a particular antiderivative? Keep in mind that $H(x) = \int_a^x f'(t)\,dt = f(x) - f(a)$, so $H(a) = 0$. Thus if we want to find an antiderivative of $h(x)$, say call it $H(x)$, such that $H(a) = A$, then we need $H(x) = \int_a^x h(u)\,du + A$. Let's find an antiderivative of $h(x) = e^{-x}\sin(5x)$ which is 10 at $x = \pi$.

$$H(x) = \int_a^x h(u)\,du + A \tag{21}$$

```
> H := int(exp(-u)*sin(5*u),u=Pi..x) + 10:
  'H(x)' = H;
```

$$H(x) = -\frac{5\,e^{-\pi}}{26} - \frac{5\,e^{-x}\cos(5x)}{26} - \frac{e^{-x}\sin(5x)}{26} + 10 \tag{22}$$

Notice that this is indeed what we want:

```
> Diff('H(x)',x) = diff(H,x);
```

$$\frac{d}{dx} H(x) = e^{-x}\sin(5x) \tag{23}$$

```
> 'H(Pi)' = simplify(subs(x=Pi,H));
```

$$H(\pi) = 10 \tag{24}$$